

```

/*Define as variáveis da fórmula.*/
DECL CODAREA, CODMATERIA, CODQUESTAO, CANDIDATO1, CANDIDATO2, RENDA, CANDDESEMPATE;

/*Define as variáveis relacionadas aos dados da questão do questionário.*/
SETVAR(CODAREA, 'CODAREA'); /*Preencha corretamente este campo*/
SETVAR(CODMATERIA, 'CODMATERIA'); /*Preencha corretamente este campo*/
SETVAR(CODQUESTAO, 'CODQUESTAO'); /*Preencha corretamente este campo*/
/*Define o valor de referência da renda.*/
SETVAR(RENDA, 10);
/*Define o valor da renda informada dos candidatos 1 e 2 no questionário socioeconômico.*/
SETVAR(CANDIDATO1, PSDNRendaSalarioQuestSocioEconomico(1, CODAREA, CODMATERIA, CODQUESTAO));
SETVAR(CANDIDATO2, PSDNRendaSalarioQuestSocioEconomico(2, CODAREA, CODMATERIA, CODQUESTAO));

/*Caso algum dos candidatos não tenha feito o questionário a função de fórmula retornará 0*/
SE CANDIDATO1 = 0 OU CANDIDATO2 = 0 ENTAO
  /*Caso nenhum dos dois tenham feito a prova, o desempate mantém o candidato 1*/
  SE CANDIDATO1 = CANDIDATO2 ENTAO
    SETVAR(CANDDESEMPATE, 1); /*Define o candidato 1*/
  SENAO
    /*Se somente o candidato 1 não tiver executado a prova, retorna o candidato 2*/
    SE CANDIDATO1 = 0 ENTAO
      SETVAR(CANDDESEMPATE, 2); /*Define o candidato 2*/
    /*Se somente o candidato 2 não tiver executado a prova, retorna o candidato 1*/
    SENAO
      SETVAR(CANDDESEMPATE, 1); /*Define o candidato 1*/
    FIMSE
  FIMSE
/*Se ambos os candidatos executaram o questionário*/
SENAO
  /*Se a renda informada dos dois candidatos for maior que o valor referência não é feita nenhuma ação,
  pois, de acordo com a legislação federal, o desempate por menor renda ocorrerá somente se as duas rendas forem
  menor que o valor de referência definido.*/
  SE CANDIDATO1 > RENDA E CANDIDATO2 > RENDA ENTAO
    SETVAR(CANDDESEMPATE, 1); /*Define o candidato 1*/
  /*Se renda de pelo menos um deles é menor que o valor de referência.*/
  SENAO
    /*Se a renda de ambos for menor que o valor de referência, porém forem iguais, será mantido a ordem*/
    SE CANDIDATO1 = CANDIDATO2 ENTAO
      SETVAR(CANDDESEMPATE, 1); /*Define o candidato 1*/
    /*Se as rendas forem diferentes.*/
    SENAO
      /*Verifica qual das duas rendas é menor*/
      SE CANDIDATO1 < CANDIDATO2 ENTAO
        SETVAR(CANDDESEMPATE, 1); /*Define o candidato 1*/
      SENAO
        SETVAR(CANDDESEMPATE, 2); /*Define o candidato 2*/
      FIMSE
    FIMSE
  FIMSE
FIMSE
/*Retorna o candidato no desempate.*/
CANDDESEMPATE

```